# MATH529L Lab01: Fourier Series

## Construction of Fourier Series

Consider a periodic function with period 2p

$$f : (-p, p) \to \mathbb{C}$$

$$f(x) = \sum_{n=-\infty}^{\infty} c_n e^{i\pi n x/p} = \sum_{n=-\infty}^{\infty} (a_n + i b_n)\left[\cos\left(\frac{\pi n x}{p}\right) + i \sin\left(\frac{\pi n x}{p}\right)\right]$$

Consider a sequence of periodic functions, starting from a step function with discontinuities inside the domain of periodicity and at the end points, to a smooth function.

### Example 1

$f(x) = 0$ for $-\pi < x < 0$, $f(x) = 1$, $0 \le x < \pi$

In[51]:=
```
p1[x_] := x  -  Floor[(x + π) / (2 π)] (2 π);
f1[x_] :=  If[p1[x] < 0, 0, 1];
```

In[53]:=
```
Plot[f1[x], {x, -5 π, 5 π}]
```

Out[53]=




In[54]:=
```
Plot[f1[x], {x, -π, π}]
```

Out[54]=




## Example 2

$f(x) = 0$ for $-\pi < x < 0$, $f(x) = x^2$, $0 \le x < \pi$

In[59]:=
```
p2[x_] := x  -  Floor[(x + π) / (2 π)] (2 π);
f2[x_] :=  If[p2[x] < 0, 0, p2[x]^2];
```

In[61]:=
```
Plot[f2[x], {x, -5 π, 5 π}]
```

Out[61]=


In[58]:=
```
Plot[f2[x], {x, -π, π}]
```

Out[58]=


## Example 3

$f(x) = \pi^2 - x^2$ for $-\pi < x < \pi$

In[64]:=
```
p3[x_] := x  -  Floor[(x + π) / (2 π)] (2 π);
f3[x_] := π^2 -  p3[x]^2;
```

In[66]:=
```
Plot[f3[x], {x, -5 π, 5 π}]
```

Out[66]=



In[67]:=
```
Plot[f3[x], {x, -π, π}]
```

Out[67]=



## Example 4

$f(x) = \sin(\cos(x)) + \cos(\sin(x))$

In[70]:=
```
f4[x_] := Sin[Cos[x]] + Cos[Sin[x]];
```

In[71]:=
```
Plot[f4[x], {x, -5 π, 5 π}]
```

Out[71]=




In[72]:=
```
Plot[f4[x], {x, -π, π}]
```

Out[72]=




# Example 5

$f(x) = 1 + \sin(x) + 10\cos(x) + 10^2\sin(2x) + 10^3\cos(3x)$

In[73]:=
```
f5[x_] := 1 + Sin[x] + 10 Cos[x] + 10^2 Sin[2 x] + 10^3 Cos[3 x];
```

In[74]:=
```
Plot[f5[x], {x, -5 π, 5 π}]
```

Out[74]=



In[75]:=
```
Plot[f5[x], {x, -π, π}]
```

Out[75]=



# Fourier Series Coefficients

Define scalar product for Fourier series with [-π,π] interval of periodicity

In[77]:=
```
ScProd[f_, g_] := 1/(2 π) Integrate[f Conjugate[g], {x, -π, π}];
```

In[85]:=
```
NScProd[f_, g_] := 1/(2 π) NIntegrate[f Conjugate[g], {x, -π, π}];
```

In[79]:=
```
Fcoef[f_, n_] := Table[ScProd[f, Exp[ⅈ k x]], {k, -n, n}];
```

In[86]:=
```
NFcoef[f_, n_] := Table[NScProd[f, Exp[ⅈ k x]], {k, -n, n}];
```

In[80]:=
```
cf1 = Fcoef[f1[x], 3]
```

Out[80]=

$$\left\{\frac{i}{3\pi}, 0, \frac{i}{\pi}, \frac{1}{2}, -\frac{i}{\pi}, 0, -\frac{i}{3\pi}\right\}$$

In[93]:=
```
Off[NIntegrate::slwcon, NIntegrate::ncvb];
cf1N = Chop[NFcoef[f1[x], 3]]
```

Out[94]=
```
{0. + 0.106103 ⅈ, 0, 0. + 0.31831 ⅈ, 0.5, 0. - 0.31831 ⅈ, 0, 0. - 0.106103 ⅈ}
```

In[82]:=
```
F[x_] = {f1[x], f2[x], f3[x], f4[x], f5[x]};
```

In[105]:=
```
nTerms = 10;
cf = Chop[Map[NFcoef[#, nTerms] &, F[x]]];
```

In[107]:=
```
RowH = Table[c_k, {k, -nTerms, nTerms}];
ColH = Table["f" <> ToString[k], {k, 1, 5}];
```

In[109]:=
```
TableForm[Transpose[cf], TableHeadings → {RowH, ColH}]
```

Out[109]//TableForm=

| | f1 | f2 | f3 | f4 | f5 |
|---|---|---|---|---|---|
| $c_{-10}$ | 0 | 0.01 − 0.15708 ⅈ | −0.02 | $2.63062\times10^{-10}$ | 0 |
| $c_{-9}$ | 0. + 0.0353678 ⅈ | −0.0123457 + 0.17366 ⅈ | 0.0246914 | $5.24925\times10^{-9}$ | 0 |
| $c_{-8}$ | 0 | 0.015625 − 0.19635 ⅈ | −0.03125 | $9.42234\times10^{-8}$ | 0 |
| $c_{-7}$ | 0. + 0.0454728 ⅈ | −0.0204082 + 0.222543 ⅈ | 0.0408163 | $-1.50233\times10^{-6}$ | 0 |
| $c_{-6}$ | 0 | 0.0277778 − 0.261799 ⅈ | −0.0555556 | 0.0000209383 | 0 |
| $c_{-5}$ | 0. + 0.063662 ⅈ | −0.04 + 0.309066 ⅈ | 0.08 | 0.000249758 | 0 |
| $c_{-4}$ | 0 | 0.0625 − 0.392699 ⅈ | −0.125 | 0.00247664 | 0 |
| $c_{-3}$ | 0. + 0.106103 ⅈ | −0.111111 + 0.50002 ⅈ | 0.222222 | −0.0195634 | 500 |
| $c_{-2}$ | 0 | 0.25 − 0.785398 ⅈ | −0.5 | 0.114903 | 0. |
| $c_{-1}$ | 0. + 0.31831 ⅈ | −1. + 0.934177 ⅈ | 2. | 0.440051 | 5. |
| $c_{0}$ | 0.5 | 1.64493 | 6.57974 | 0.765198 | 1. |
| $c_{1}$ | 0. − 0.31831 ⅈ | −1. − 0.934177 ⅈ | 2. | 0.440051 | 5. |
| $c_{2}$ | 0 | 0.25 + 0.785398 ⅈ | −0.5 | 0.114903 | 0. |
| $c_{3}$ | 0. − 0.106103 ⅈ | −0.111111 − 0.50002 ⅈ | 0.222222 | −0.0195634 | 500 |
| $c_{4}$ | 0 | 0.0625 + 0.392699 ⅈ | −0.125 | 0.00247664 | 0 |
| $c_{5}$ | 0. − 0.063662 ⅈ | −0.04 − 0.309066 ⅈ | 0.08 | 0.000249758 | 0 |
| $c_{6}$ | 0 | 0.0277778 + 0.261799 ⅈ | −0.0555556 | 0.0000209383 | 0 |
| $c_{7}$ | 0. − 0.0454728 ⅈ | −0.0204082 − 0.222543 ⅈ | 0.0408163 | $-1.50233\times10^{-6}$ | 0 |
| $c_{8}$ | 0 | 0.015625 + 0.19635 ⅈ | −0.03125 | $9.42234\times10^{-8}$ | 0 |
| $c_{9}$ | 0. − 0.0353678 ⅈ | −0.0123457 − 0.17366 ⅈ | 0.0246914 | $5.24925\times10^{-9}$ | 0 |
| $c_{10}$ | 0 | 0.01 + 0.15708 ⅈ | −0.02 | $2.63062\times10^{-10}$ | 0 |

In[113]:=
```
Table[(k - 1 - (Length[cf1] - 1) / 2), {k, 1, Length[cf1]}]
```

Out[113]=
```
{-3, -2, -1, 0, 1, 2, 3}
```

In[118]:=
```
fct[c_, k_] := (k - 1 - (Length[c] - 1) / 2)
```

In[119]:=
```
Fseries[c_] := Sum[ c[[k]] Exp[ⅈ fct[c, k] x] , {k, 1, Length[c]}]
```

In[110]:=
```
f5[x]
```

Out[110]=
```
1 + 10 Cos[x] + 1000 Cos[3 x] + Sin[x] + 100 Sin[2 x]
```

In[125]:=
```
Chop[ComplexExpand[Simplify[Fseries[cf[[5]]]]]]
```

Out[125]=
```
1. + 10. Cos[x] + 1000. Cos[3 x] + 1. Sin[x] + 100. Sin[2 x]
```

In[126]:=
```
fseries = Map[Chop[ComplexExpand[Simplify[Fseries[#]]]] &, cf]
```

Out[126]=

```
{0.5 + 0.63662 Sin[x] + 0.212207 Sin[3 x] +
   0.127324 Sin[5 x] + 0.0909457 Sin[7 x] + 0.0707355 Sin[9 x],
 1.64493 - 2. Cos[x] + 0.5 Cos[2 x] - 0.222222 Cos[3 x] + 0.125 Cos[4 x] - 0.08 Cos[5 x] +
   0.0555556 Cos[6 x] - 0.0408163 Cos[7 x] + 0.03125 Cos[8 x] -
   0.0246914 Cos[9 x] + 0.02 Cos[10 x] + 1.86835 Sin[x] - 1.5708 Sin[2 x] +
   1.00004 Sin[3 x] - 0.785398 Sin[4 x] + 0.618133 Sin[5 x] - 0.523599 Sin[6 x] +
   0.445087 Sin[7 x] - 0.392699 Sin[8 x] + 0.347319 Sin[9 x] - 0.314159 Sin[10 x],
 6.57974 + 4. Cos[x] - 1. Cos[2 x] + 0.444444 Cos[3 x] - 0.25 Cos[4 x] + 0.16 Cos[5 x] -
   0.111111 Cos[6 x] + 0.0816327 Cos[7 x] - 0.0625 Cos[8 x] + 0.0493827 Cos[9 x] - 0.04 Cos[10 x],
 0.765198 + 0.880101 Cos[x] + 0.229807 Cos[2 x] - 0.0391267 Cos[3 x] + 0.00495328 Cos[4 x] +
   0.000499515 Cos[5 x] + 0.0000418767 Cos[6 x] - 3.00465 × 10^-6 Cos[7 x] +
   1.88447 × 10^-7 Cos[8 x] + 1.04985 × 10^-8 Cos[9 x] + 5.26123 × 10^-10 Cos[10 x],
 1. + 10. Cos[x] + 1000. Cos[3 x] + 1. Sin[x] + 100. Sin[2 x]}
```

# Convergence of Fourier Series

In[138]:=

```
cmpplt[k_] := Plot[{fseries[[k]], F[x][[k]]}, {x, -π, π}, Frame → True, GridLines → Automatic,
   PlotStyle → {Red, Blue}, PlotLabel → "Convergence of Fourier Series"];
```

In[140]:=

```
Table[cmpplt[k], {k, 1, 5}]
```

Out[140]=




# Questions to Investigate

Gather all the intermediate steps above into a single function producing a plot of the approxima -

tion of f given n the number of terms in the Fourier series

In[141]:=
```
FConvPlot[f_, n_] := Module[{c, F, plt},
  c = NFcoef[f, n];
  F = Chop[ComplexExpand[Simplify[Fseries[c]]]];
  plt = Plot[{F, f}, {x, -π, π}, Frame → True, GridLines → Automatic,
    PlotStyle → {Red, Blue}, PlotLabel → "Convergence of Fourier Series"];
  Return[plt]
 ]
```

Study convergence of Fourier series of a step function

In[144]:=
```
Manipulate[FConvPlot[f1[x], k], {k, 1, 11, 2}]
```

Out[144]=




# Question 1

Construct an animation of the convergence of a Fourier series for f1, f2, f3

# Question 2

Construct a plot of $| c_n |$, the absolute value of the Fourier series coefficients, as a function of the index n using linear axes. Repeat using a logarithmic axis, i.e., $\log | c_n |$.

## Question 3

Construct f6(x)=f4(x)+f1(f5(x)). Present a convergence study (n=5,10,15,20).